

点阵图形液晶说明书

Model: SYG2401281

北京信通时代科技有限公司

地址：北京市海淀区中关村南大街 30 号 420 室

电话：010-62139879，62135055，66801148

传真：010-62135055 邮编：100081

E-mail：syntop@syntoplcd.com

网址：<http://www.syntoplcd.com>

目 录

一、概述-----	2
1.1 模块基本类型-----	2
1.2 模块尺寸-----	2
1.3 模块介绍-----	2
二、外形尺寸-----	2
三、模块的外部接口-----	4
四、T6963C 液晶显示控制器功能说明及指令表-----	4
五、电气参数-----	5
5.1 极限参数-----	5
5.2 直流特性参数-----	5
5.3 M C U总线交流参数-----	6
六、指令功能描述-----	6
6.1 状态判断-----	6
6.2 带参数命令设置-----	7
6.3 命令介绍-----	7
6.3.1 地址指针设置-----	7
6.3.2 光标地址设置-----	7
6.3.3 CGRAM 偏置地址设置-----	7
6.3.4 读写显缓地址设置-----	8
6.3.5 显示区域设置-----	8
6.3.6 文本显示区首地址及宽度设置-----	8
6.3.7 图形显示区首地址及宽度设置-----	8
6.3.8 显示方式设置-----	9
6.4 数据自动读写-----	9
6.5 数据读写一次-----	9
6.6 屏读一字节-----	9
6.7 屏读一行-----	10
6.8 显示缓存数据位操作-----	10
七、应用举例-----	10

一、概述

SYG2401281 有 SYG2401281-SFLYS (H) 6 (12) V (为黄绿屏、普通 LED 底背光、常温 (宽温) 单 5V 供电下 (上) 视角液晶模块) SYG2401281-SFDYS (H) 6 (12) V (为黄绿屏、普通 LED 侧背光、常温 (宽温) 单 5V 供电下 (上) 视角液晶模块) SYG2401281-SFDWS (H) 6 (12) V (为黄绿屏、白色 LED 背光、常温 (宽温) 单 5V 供电下 (上) 视角液晶模块) SYG2401281-BMDWS (H) 6 (12) V (为蓝屏负显、白色 LED 背光、常温 (宽温) 单 5V 供电下 (上) 视角液晶模块) 等几种型号。

1.1 模块基本类型：

- 1、LCD 类型：STN 黄绿模正显示、蓝膜负显及 FSTN 正、负显示等；
- 2、驱动方式：1/128 Duty，1/12 Bias；
- 3、显示视角：6：00、12：00
- 4、背光：LED 普通背光、LED 白光。


1.2 模块尺寸：

类 别	尺 寸	单 位
模块外形尺寸	144.0×104.0×13.7	mm
视域尺寸	114.0×64.0	mm
点阵数	240×128	点
点尺寸	0.40×0.40	mm
点间距	0.45×0.45	mm

1.3 模块介绍：

- 1、单 5V 供电方式，自带 DC-DC 负压转换。
- 2、内含 T6963C 控制器，可直接与单片机 (MCS-51 系列及其它系列) 接口。
- 3、带 32K 显示缓冲数据存储器，同时可存储八屏多图形显示信息。

二、外形尺寸



三、模块外部接口定义

序 号	名称	电 平	功能描述
1	FG	0V	铁框地
2	VSS	0V	信号地
3	VDD	5V	工作电压
4	VO	VOUT V0 VDD	对比度调节输入电压
5	/WR	L	写信号
6	/RD	L	读信号
7	/CE	L	片选信号
8	C/D	H/L	指令/数据选择 (H: 指令; L: 数据)
9	/RESET	L	复位 (模块已带上电自动复位电路)
10	DB0	H//L	数据线 0 (三态数据总路线)
11	DB1	H/L	数据线 1
12	DB2	H/L	数据线 2
13	DB3	H/L	数据线 3
14	DB4	H/L	数据线 4
15	DB5	H/L	数据线 5
16	DB6	H/L	数据线 6
17	DB7	H/L	数据线 7
18	FS	H/L	字体选择 (H: 6×8 点; L: 8×8 点)
19	VOUT	输出-15V	负电压输出。(双电源供电时为负电压输入)
20	LEDA	5V	LED 背光输入 (或 EL 背光电源输入)
21	LEDK	0V	LED 背光输入 (或 EL 背光电源输入)
22	NC	--	空

四、T6963C 液晶显示控制器功能说明及指令表

T6963C 是日本东芝公司专门为中等规模 LCD 模块设计的一款控制器,它通过外部 MCU 方便地实现对 LCD 驱动器和显示缓存的管理。其特点为 8 位 80 或 Z80 系列总线,内部有 128 个常用字符表,可管理外部扩展显示缓存 64KB (本模块为 32KB),并具有丰富的指令供 MCU 实现对 LCD 显示屏幕的操作与编辑。

指令表

命令	命令码	参数 D1	参数 D2	功能
地址指针设置	00100001 (21H) 00100010 (22H) 00100100 (24H)	X 横向地址 偏置地址 低 8 位地址	Y 垂直地址 00H 高 8 位地址	光标地址设置 CGRAM 偏置地址设置 读写显缓地址设置
显示区域设置	01000000 (40H) 01000001 (41H) 01000010 (42H) 01000011 (43H)	低 8 位地址 每行字符数 低 8 位地址 每行字节数	高 8 位地址 00H 高 8 位地址 00H	文本显示区首地址 文本显示区宽度 图形显示区首地址 图形显示区宽度
显示方式设置	10000000 (80H) 10000001 (81H) 10000011 (83H) 10000100 (84H)	-- -- -- --	-- -- -- --	文本与图形逻辑"或"合成显示 文本与图形逻辑"异或"合成显示 文本与图形逻辑"与"合成显示 文本显示特征以双字节表示
显示状态设置	10010000 (90H) 10010010 (92H) 10010011 (93H) 10010100 (94H) 10011000 (98H) 10011100 (9CH)	-- -- -- -- -- --	-- -- -- -- -- --	关所有显示 光标显示但不闪 光标闪动显示 文本显示,图形关闭 文本关闭,图形显示 文本和图形都显示
光标大小设置	10100000 (A0H) 10100001 (A1H) 10100010 (A2H) 10100011 (A3H) 10100100 (A4H) 10100101 (A5H)	-- -- -- -- -- --	-- -- -- -- -- --	1 行八点光标 2 行八点光标 3 行八点光标 4 行八点光标 5 行八点光标 6 行八点光标

	10100110 (A6H) 10100111 (A7H)			7 行八点光标 8 行八点光标
进入/退出显示 数据自动读/写 方式设置	10110000 (B0H) 10110001 (B1H) 10110010 (B2H) 10110011 (B3H)			进入显示数据自动写方式 进入显示数据自动读方式 退出自动读/写方式 退出自动读/写方式
进入显示数据一 次读/写方式设 置	11000000 (C0H) 11000001 (C1H) 11000010 (C2H) 11000011 (C3H) 11000100 (C4H) 11000101 (C5H)	数据 -- 数据 -- 数据 --		写一字节数据, 地址指针加一 读一字节数据, 地址指针加一 写一字节数据, 地址指针减一 读一字节数据, 地址指针减一 写一字节数据, 地址指针不变 读一字节数据, 地址指针不变
屏读一字节	11100000 (E0H)			从当前地址指针 (在图形区内) 读一字节屏幕 显示数据
屏读拷贝(一行)	11101000 (E8H)			从当前地址指针 (在图形区内) 读一行屏幕显 示数据并写回
显示数据位操作 设置	11110XXX (F0-F7H) 11111XXX (F8-FFH) 1111X000 1111X001 1111X010 1111X011 1111X100 1111X101 1111X110 1111X111	-- -- -- -- -- -- -- -- -- --	-- -- -- -- -- -- -- -- -- --	位清零 位置位 设位地址 Bit 0 (LSB) 设位地址 Bit 1 设位地址 Bit 2 设位地址 Bit 3 设位地址 Bit 4 设位地址 Bit 5 设位地址 Bit 6 设位地址 Bit 7 (MSB)

五、电气参数

5.1 极限参数


Item	Symbol	Condition	Min.	Typ.	Max.	Unit
Supply Voltage(logic)	VDD-VSS	--	-0.3	--	7	V
Supply Voltage(LCDDrive)	VDD-V0	--	0	--	15	V
Input	VI	--	-0.3	--	VDD+0.3	V
Operating Temperature	Topr	--	0(-20)	--	+50(+70)	
Storage Temperature	Tstg	--	-20(-30)	--	+60(+80)	

5.2 直流特性参数

Item	Symbol	Condition		Standard Value			Unit
				Min.	Typ.	Max.	
Supply voltage(Logic)	VDD-VSS	-		4.75	5	5.25	V
Supply current	IDD	-		--	100.0	150.0	mA
	IO	-		--	1.8	2.5	mA
Ied Backlight current	Hed	-		--	150	--	mA
Input high voltage	VIH	High level		0.7VDD	--	VDD	V
Input low voltage	VIL	Low level		0	--	0.3VDD	V
Supply voltage for LCD	VDD-V0	Ta=0		14.2	14.5	14.8	V
				13.3	13.6	13.9	V
				12.3	12.6	12.9	V
Contrast Ratio	CR			--	4	--	--
Viewing Angle	--	CR 2	θ	-10	--	20	deg
			θ	60	--	120	deg
Response Time (rise)	Tr	Note 1	Ta=25	--	130	200	ms
Response Time (delay)	Td	Note 2	Ta=25	--	150	230	ms

5.3 M C U总线交流参数

Item	Symbol	Min.	Max.	Unit
C/D set up time	Tcds	100	--	ns
C/D hold time	Tcdh	10	--	ns
CE,RD,WR pulse width	Tce,Trd,Twr	80	--	ns
DATA set up time	Tds	80	--	ns
DATA hold time	Tdh	40	--	ns
Access time	Tacc	--	150	ns
Output hold time	Toh	10	50	ns



六、指令功能描述

6.1 状态判断

无论是向 T 6 9 6 3 C 读写数据还是写入命令，都必需判断忙状态。读忙状态满足以下条件：

/RD L
/WR H
/CE L
C/D H
D0 D7 状态字

T6963C 状态字定义如下：


MSB

LSB

STA7	STA6	STA5	STA4	STA3	STA2	STA1	STA0
D7	D6	D5	D4	D3	D2	D1	D0

STA0	指令读写状态	0：忙 1：闲
STA1	数据读写状态	0：忙 1：闲
STA2	数据自动读状态	0：忙 1：闲
STA3	数据自动写状态	0：忙 1：闲
STA4	未用	
STA5	控制器运行检测可能性	0：不能 1：可能
STA6	屏读/屏拷贝出错状态	0：对 1：错
STA7	闪烁状态检测	0：关 1：开

说明 1、STA0 和 STA1 在大多数命令和数据传送前必需在同一时刻判断,否则可能会出错；
2、在数据自动读写时判断 STA2 和 STA3；
3、在屏读 / 屏拷贝时判断 STA6；
4、STA5 和 STA7 为厂家测试时用。





6.2 带参数命令设置

带参数命令设置先送参数再送命令(注意最多只有两个参数)。

一个参数的命令

两个参数的命令



6.3 命令介绍

6.3.1 地址指针设置

命令码	HEX	功能	参数 D1	参数 D2
00100001	21H	光标地址设置	X 横向地址	Y 纵向地址
00100010	22H	CGRAM 偏置地址设置	偏置地址	00H
00100100	24H	读写显缓地址设置	低 8 位地址	高 8 位地址

6.3.2 光标地址设置

此命令用来改变光标的当前位置。

X 横向地址 00H -- 4FH(低 7 位有效)

Y 纵向地址 00H -- 1FH(低 5 位有效)

6.3.3 CGRAM 偏置地址设置

CGRAM 偏置地址是用来确定 CGRAM 16 位地址的高 5 位,此时 16 位地址定义如下：

T6963C 的 16 位地址线如下：

MSB
LSB

AD15	AD14	AD13	AD12	AD11	AD10	AD9	AD8	AD7	AD6	AD5	AD4	AD3	AD2	AD1	AD0
------	------	------	------	------	------	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

高 5 位 (ad15 - ad11) 为 CGRAM 偏置地址，次 8 位(ad10 - ad3)为字符代码，低 3 位 (ad2 - ad0) 定义该字符的 8 行地址 0—7，用户可用此命令在 CGRAM 中来定义自己的特殊字符。

6.3.4 读写显缓地址设置

读写显缓地址设置用来定义当前读写操作地址。

6.3.5 显示区域设置

该命令用来定义显示区域的大小。

命令码	HEX	功能	参数 D1	参数 D2
01000000	40H	文本显示区首地址	低 8 位地址	高 8 位地址
01000001	41H	文本显示区宽度	每行字符数	00H
01000010	42H	图形显示区首地址	低 8 位地址	高 8 位地址
01000011	43H	图形显示区宽度	每行字节数	00H

6.3.6 文本显示区首地址及宽度设置

该首地址为屏幕的左上角的第一个字符在显示缓存中的地址，注意该地址是一个以 8X8 点阵单位的地址，宽度为每行的字符数。

例如：

文本显示区首地址：0000H
文本显示区宽度：001EH（每字符行显示 30 个 8X8 的字符）
显示与地址的对应关系：

屏左上角

0000H	0001H		001CH	001DH	第一字符行
001EH	001FH		003AH	003BH	第二字符行
:	:	:	:	:	:
:	:	:	:	:	:

屏右下角

6.3.7 图形显示区首地址及宽度设置

该首地址为屏幕的左上角的第一个横向字节在显示缓存中的地址,宽度为每行的字节数。

例如：

图形显示区首地址：0000H
图形显示区宽度：001EH（每点阵行显示 30 个字节的图形信息）
显示与地址的对应关系：

屏左上角

0000H	0001H		001CH	001DH	第一点阵行
001EH	001FH		003AH	003BH	第二点阵行
:	:	:	:	:	:
:	:	:	:	:	:

屏右下角


6.3.8 显示方式设置

该命令设置图形与文本以何种逻辑关系叠加显示和文本属性显示，逻辑关系有“或”、“与”和“异或”三种，显示效果如下图所示。内部字符表代码范围为 00H-7FH。

当设为文本属性显示后，图形显示区将转换成文本属性区，文本属性代码由一字节的低 4 位组成：

X	X	X	X	D3	D2	D1	D0	X：无关
---	---	---	---	----	----	----	----	------

D3	D2	D1	D0	显示效果
0	0	0	0	正向显示
0	1	0	1	反白显示
0	0	1	1	禁止正向显示
0	1	0	0	禁止反向显示
1	0	0	0	正向闪烁显示
1	1	0	1	反向闪烁显示
1	0	1	1	禁止正向闪烁显示
1	1	0	0	禁止反向闪烁显示



6.4 数据自动读写

命令码	HEX	功能	参数
10110000	B0H	进入数据自动写	--
10110001	B1H	进入数据自动读	--
10110010(1)	B2H/B3H	退出自动读写	--

该命令将进入或退出数据自动读写操作，在数据自动读写操作中，MCU 可连续将显示数据写入显缓中或读出，在每次读或写操作后，显示地址自动加一，注意这时忙状态要判断 STA2 和 STA3，在数据自动读写操作未退出前写入其他命令是无效的。

6.5 数据读写一次

命令码	HEX	功能	参数
11000000	C0H	写入一个数据显示地址加一	要写入的数据 Data
11000001	C1H	读出一个数据显示地址加一	--
11000010	C2H	写入一个数据显示地址减一	要写入的数据 Data
11000011	C3H	读出一个数据显示地址减一	--
11000100	C4H	写入一个数据显示地址不变	要写入的数据 Data
11000101	C5H	读出一个数据显示地址不变	--

该命令是一次数据读写操作,在每次读写数据操作后,显示地址都要按指令码设置修正。

6.6 屏读一字节

命令码	HEX	功能	参数
11100000	E0H	从屏幕上读出一字节	--

该命令是一次屏幕数据读操作，屏幕数据可能是文本与图形按所设逻辑合成显示内容，在读数之前除判断 STA1 和 STA2 之外，还需判断 STA6 后才可读到正确的数据。

6.7 屏读一行

命令码	HEX	功能	参数
11101000	E8H	从屏幕上读出一行显示数据并写回	--

该命令是一行屏幕数据读并写回操作，屏幕数据可能是文本与图形合成显示内容，判断该命令是否完成除判 STA1 和 STA2 之外，还需判断 STA6 后才可写入别的命令。


6.8 显示缓存数据位操作

命令码	HEX	功能	参数
11110000	F0H	清零第 0 位	--
11110001	F1H	清零第 1 位	--
11110010	F2H	清零第 2 位	--
11110011	F3H	清零第 3 位	--
11110100	F4H	清零第 4 位	--
11110101	F5H	清零第 5 位	--
11110110	F6H	清零第 6 位	--
11110111	F7H	清零第 7 位	--
11111000	F8H	置位第 0 位	--
11111001	F9H	置位第 1 位	--
11111010	FAH	置位第 2 位	--
11111011	FBH	置位第 3 位	--
11111100	FCH	置位第 4 位	--
11111101	FDH	置位第 5 位	--
11111110	FEH	置位第 6 位	--
11111111	FFH	置位第 7 位	--

该命令可对显示地址指针所指的显缓存数据进行位操作，使用该命令使屏幕编在图形方式下达到一点的精度。

七、应用举例

SYG2401281 与单片机 AT89C52 的硬件接口如图下图所示：



说明：

- 1. 虚框内为外供负电压调节对比度方法，若模块内已内带负电压，VO 及 VOUT 空接，复位使用模块内上电

复位。若用 LED 背光，LEDA 接背光电源+5V(200mA)。

2. 片选信号/CE 接 CPU 的 P2.7(A15)，则当 A15=0 时模块选定。
3. 命令/数据选择 C/D 接 CPU 的 P2.0(A8)，则当 A8=1 时为命令，A8=0 时为数据。
4. 由 2，3 得到 LCD 模块典型命令口地址为 0100H，典型数据口地址为 0000H。

附录：使 LCD 显示一屏图形

1. 用户事先要做的是在 WINDOWS 的画笔中(点击左下角开始 程序 附件 画图)对 LCD 显示屏在电脑上
进行预编辑，完成后取名 DEMO.BMP 并保存。

2. 图形制作方法：

- A、进入 WINDOWS 画笔，在“图象” “属性” “宽度=240，高度=128，象素 黑白” “确定”。
- B、利用画笔工具对屏幕先在电脑上编辑，完成后再“图象” “翻转/旋转” “垂直翻转” “确定”。
- C、“图象” “反色”，B，C 是为了使 BMP 数据存放格式与 LCD 所需一致，减轻 CPU 处理时间。

3. CPU 内部 ROM 空间安排，将以下源文件 DEMO.ASM 编译生成 DEMO.HEX 文件。在 CPU 烧录时先从
ROM 地址 0 调入 DEMO.HEX(格式为十六进制 HEX)，再从 ROM 地址 0800H 处调入 DEMO.BMP(格式为二进
制 BIN)，将该合并成的文件烧录到 AT89C55 中即可。

4. 现让屏幕先显示 DEMO.BMP，观察其效果。注意字体选择 FS(PIN18)一定要接低，否则所显图形会错位。

5. 以下为用 MCS-51 汇编语言编写的 DEMO 源程序，该程序中的子程经过调试可直接引用。

```

;-----
PORTD      EQU      0000H      ; 模块数据口地址(A15=0.A8=0)
PORTC      EQU      0100H      ; 模块命令口地址(A15=0.A8=1)
BMP_DEMO   EQU      0800H      ; 定义位图存放首地址
;-----

                      ORG      0000H
                      LJMP     MAIN
                      ORG      0030H

MAIN:

                      MOV      SP,#60H          ; 设堆栈
                      LCALL    DEL_10MS        ; 模块上电复位
                      LCALL    LCD_INT         ; 模块初始化
                      MOV      DPTR,#BMP_DEMO
                      LCALL    SCREEN          ; 显示一正屏位图
                      AJMP     $               ; 观察效果

;-----
;                      延时子程
;-----
DEL_10MS:
                      MOV      R6,#10

DL2:
                      MOV      R7,#200

DL1:
                      NOP
                      NOP
                      NOP
                      DJNZ     R7,DL1
                      DJNZ     R6,DL2
                      RET

;-----
; 写两个参数 (参数 1 在 DPL 中, 参数 2 在 DPH 中)
;-----
WD2:
                      PUSH     DPL
                      PUSH     DPH

```

```

PUSH    ACC
MOV      A,DPL
LCALL    WD1
MOV      A,DPH
LCALL    WD1
POP      ACC
POP      DPH
POP      DPL
RET

```

```

;-----
; 写一个参数 ( 参数在 A 中 )
;-----

```

WD1:

```

PUSH    DPL
PUSH    DPH
PUSH    ACC
PUSH    ACC
LCALL    CH_STA1
POP      ACC
MOV      DPTR,#PORTD
MOVBX    @DPTR,A
POP      ACC
POP      DPH
POP      DPL
RET

```

```

;-----
; 自动方式下写一个显示数据 ( 数据在 A 中 )
;-----

```

AWD:

```

PUSH    DPL
PUSH    DPH
PUSH    ACC
PUSH    ACC
LCALL    CH_STA3
POP      ACC
MOV      DPTR,#PORTD
MOVBX    @DPTR,A
POP      ACC
POP      DPH
POP      DPL
RET

```

```

;-----
; 写一个命令 ( 命令字在 A 中 )
;-----

```

WC:

```

PUSH    DPL
PUSH    DPH
PUSH    ACC
PUSH    ACC
LCALL    CH_STA1
POP      ACC
MOV      DPTR,#PORTC
MOVBX    @DPTR,A
POP      ACC
POP      DPH
POP      DPL
RET

```

```

;-----
;                               判断状态 0,1
;-----
CH_STA1:
        PUSH    DPL
        PUSH    DPH
        PUSH    ACC
        MOV     DPTR,#PORTC

CH_1:
        MOVX    A,@DPTR
        ANL     A,#03H
        CJNE    A,#03H,CH_1
        POP     ACC
        POP     DPH
        POP     DPL
        RET

;-----
;                               判断状态 3
;-----
CH_STA3:
        PUSH    DPL
        PUSH    DPH
        PUSH    ACC
        MOV     DPTR,#PORTC

CH_3:
        MOVX    A,@DPTR
        ANL     A,#08H
        CJNE    A,#08H,CH_3
        POP     ACC
        POP     DPH
        POP     DPL
        RET

;-----
;                               LCD 初始化
;-----
LCD_INT:
        MOV     DPTR,#00H           ; 图形显示区首地址为 00H ;
        LCALL    WD2                 ; 写入两个参数 ;
        MOV     A,#42H               ;
        LCALL    WC                 ; 写入设置图形显示区首地址命令 42H ;
        MOV     DPTR,#001EH         ; 图形显示区宽度为 30=240/8 ;
        LCALL    WD2                 ; 写入两个参数 ;
        MOV     A,#43H               ;
        LCALL    WC                 ; 写入设置图形显示区宽度命令 43H ;
        MOV     A,#80h               ;
        LCALL    WC                 ; 写入设置文本与图形逻辑 “ 或 ” 合成显示 ;
        MOV     A,#98H               ;
        LCALL    WC                 ; 写入设置显示方式仅为图形显示 ;
        RET

;-----
; 显示一幅 240X128 的黑白位图,位图在 ROM 中存放首地址在 DPTR 中
;-----
SCREEN:
        PUSH    DPL
        PUSH    DPH

```

```

MOV    DPTR,#00H    ;设置显缓写地址
LCALL  WD2
MOV    A,#24H
LCALL  WC
MOV    A,#0B0H      ;设置进入显示数据自动写模式
LCALL  WC
POP    DPH
POP    DPL
MOV    R7,#128      ;共 128 行
GRAPH:
MOV    R6,#30        ;每行 30 个字节
GRAPH1:
MOV    A,#3EH        ;ROM 中显示数据存放地址偏移量
MOVC   A,@A+DPTR     ;从 ROM 中取显示数据
LCALL  AWD            ;自动写一个显示数据，显缓地址指针自动加一
INC    DPTR          ;调整 ROM 数据指针
DJNZ   R6,GRAPH1     ;一行未写完继续写
INC    DPTR          ;调整 ROM 数据指针
INC    DPTR
DJNZ   R7,GRAPH      ;128 行未写完继续写
LCALL  CH_STA3       ;检测状态 3
MOV    A,#0B2H      ;
LCALL  WC            ;退出自动写模式
RET

```